
The RFQ is driven at eight points along the structure.  Each of the four modules 
incorporates two drive loops in opposing quadrants, with the same quadrants driven in 
each of the four modules.  The nominal peak power supplied by each drive loop is less 
than 100 kW, for a total of 800 kW drive power, including beam loading.  Each loop is 
matched to a line impedance of 50 ohms.

An r.f. power distribution system delivers the power, supplied by a single klystron, with a 
circulator, from a 9 inch, 50 ohm coaxial line to the power distribution system, which, in 
turn, delivers power to each of the eight drive loops.  This note analyzes the circuit 
topology of the power distribution system.

The power distribution system comprises a cascade of three sets of 3-db power splitters, 
each of a different topology.  The input power is split into two channels, each of which is 
again split into two channels, and finally, each finally split in four ring hybrids into the 
final eight channels.

Figure 1 shows the first two sets of dividers.

The power from the klystron is introduced into 50 ohm port A, and split into the two 
branches at point B.  The line from B to C and B to C’ is a 71 ohm transformer, 
converting the 50 ohm impedance of line CD and C’D’ each to 100 ohms at point B.  
These transformers are each 3λ/4 long in electrical length, rather than the usual λ/4.  The 
electrical distance from C to D and C’ to D’ is also 3λ/4, although this distance is 
non-critical, as it is a matched 50 ohms part of the line.  The parallel combination of both 
100 ohms load at point B results in a 50 ohm load, which is seen from point B, driven 
from A.  There is no isolation between the left and right branches of the circuit in Figure 
1.

At point D the two ports E and F present themselves as 100 ohm loads, similarly at point 
D’.  How this is done is described after describing the ring hybrid, shown in Figure 2.

The ring hybrid provides one input at port J, and four outputs at ports K, L, K’ and L’.  
The electrical circumference of the hybrid is 6λ/4.  Any two ports separated by an even 
multiple of λ/4 are isolated from each other, and any two ports separated by an odd 



multiple of λ/4 are coupled.  Therefore, power introduced at port J is coupled to ports K 
and K’, which are isolated from each other, and ports L and L’ are isolated from J.  
Power reflected from K or K’ will be coupled to all of J, L and L’.

The ports K and K’ are connected to two drive ports of the RFQ, and ports L and L’ are 
connected to 50 ohms loads.  Power reflected back from the RFQ to ports K and K’ will 
be directed to ports J, L and L’.  Power reflected to L and L’ will be dissipated in the 50 
ohm loads, and power reflected to J will be dissipated in the 50 ohm connected to the 
circulator following the klystron.

Normally, for a ring hybrid that matches 50 ohms on all its ports has a 71 ohm impedance 
of the ring itself.  However, in this case, the ring is built of 50 ohms components, and the 
input impedance at point M with all outputs terminated in 50 ohms is 25 ohms.  In this 
way, conventional 50 ohm components can be used in the ring hybrid.

Port J of four ring hybrids are connected to ports E, F, E’ and F’ of the main distribution 
trunk.  As mentioned above, points D and D’ must each see two 100 ohm loads in parallel 
to result in a 50 ohm load at points D and D’.  (In addition, it is assumed that 
quarter-wave stubs short points D and D’ to ground, to provide mechanical support for 
the center conductor of the main line.  If these shorts do not exist, the system will operate 
in the same manner.)

The 25 ohm load impedance of point M in each ring hybrid is transformed to a 100 ohm 
load at point D by a quarter-wave 50 ohm transformer between point M and point D, 
where the port J of each of the four ring hybrids is attached to ports E, F, E’ and F’ of the 
main distribution trunk.  Again, the quarter-wave transformer is constructed of 50 ohm 
line, so conventional 50 ohm components may be used, as in the ring hybrid.

For the special case where only one module of the RFQ is to be driven with one ring 
hybrid, the special transformer in Figure 3 is used.  The point M in the ring hybrid 
represents a 25 ohm load.  Connecting the special transformer port Q’ to J results in a 
quarter-wave transformer of 50 ohm line transforming the 25 ohm impedance of point M 
to 100 ohm at point P’.  Then, a quarter-wave transformer with a 71 ohm impedance from 
point P’ to P transforms the 100 ohm impedance at point P’ to 50 ohms at point P.  The 
line from point P to Q has a 50 ohm impedance, and thus propagates to 50 ohm 
impedance at P to Q, and to the drive line preceding it. The connectors at each end are 
conventional 50 ohm components.

The system is very cleverly designed:  all components use conventional 50 ohm line and 
connectors, and only one short section from C to C’ is of an impedance other than 50 
ohms.  The ring hybrids provide locations for local 50 ohm loads for reflected power 
from the RFQ drive ports.  Some power reflected back from the RFQ will be reflected 
back toward the klystron and be dissipated in the circulator system.

The following table at the end lists the Spice3.4 input data, using a transmission line 
model of the power distribution system.



Full 8-way power distribution manifold system

*
*  RFQ simulated by tuned circuit with Q0 = 6600
*
*          T3:    50ohms, 1/4 wavel short
*                   |
*      hybA<-----node(b13) ------>hybB
*                   |
*          T2:    50ohms, 3/4 wavel
*                   |
*          T1:    71ohms, 3/4 wavel
*                   |
*   gen--50ohms--node(bin)
*                   |
*          T4:    71ohms, 3/4 wavel
*                   |
*          T5:    50ohms, 3/4 wavel
*                   |
*      hybC<-----node(b23) ------>hybD
*                   |
*          T6:    50ohms, 1/4 wavel short
*
*
*                     RA1       LA1
*  3 db ring hybrid:    \       /
*  all 50 ohm            f1----l1
*                       /        \
*                      /          \
*      in--1/4 wave---no           |
*                      \          /
*                       \        /
*                        f2----l2
*                       /        \
*                     RA2        LA2

.ac lin 201 400.00Meg 405.0Meg

Vin  bgen 0 ac 5.66 dc 0
Rgen bgen bin 50

* 3 db hybrid subcircuit
.subckt hybrid in f1 f2 l1 l2
T10 in 0 no 0 z0=50 f=402.5Meg nl=0.25
T11 no 0 f1 0 z0=50 f=402.5Meg nl=0.25
T12 f1 0 l1 0 z0=50 f=402.5Meg nl=0.25
T13 l1 0 l2 0 z0=50 f=402.5Meg nl=0.50
T14 l2 0 f2 0 z0=50 f=402.5Meg nl=0.25
T15 f2 0 no 0 z0=50 f=402.5Meg nl=0.25
.ends

* from drive point to one hybrid pair
* 3/4 wave xfmr, followed by 3/4 wave and a 1/4 wave short
T1 bin 0 b12 0 z0=70.71 f=402.5Meg nl=0.75
T2 b12 0 b13 0 z0=50    f=402.5Meg nl=0.75
T3 b13 0 0   0 z0=50    f=402.5Meg nl=0.25

* from drive point to other hybrid pair
T4 bin 0 b22 0 z0=70.71 f=402.5Meg nl=0.75
T5 b22 0 b23 0 z0=50    f=402.5Meg nl=0.75
T6 b23 0 0   0 z0=50    f=402.5Meg nl=0.25

* RA are RFQ loads, RAL are 50 ohm absorbers
LA1 ra1 0 0.374pH
LA2 ra2 0 0.374pH
RA3 la1 0 50



RA4 la2 0 50

LB1 rb1 0 0.374pH
LB2 rb2 0 0.374pH
RB3 lb1 0 50
RB4 lb2 0 50

LC1 rc1 0 0.374pH
LC2 rc2 0 0.374pH
RC3 lc1 0 50
RC4 lc2 0 50

LD1 rd1 0 0.374pH
LD2 rd2 0 0.374pH
RD3 ld1 0 50
RD4 ld2 0 50

* resonant RFQ load, Q0 = 6600
* has 8 loops coupled, each 50 ohms back impedance
* loaded Q is 3300
Lrfq rfq 0 0.374pH
Crfq rfq 0 0.4180583uF
Rrfq rfq 0 6.25

* mutual couplings
k1 LA1 Lrfq .9999
k2 LA2 Lrfq .9999
k3 LB1 Lrfq .9999
k4 LB2 Lrfq .9999
k5 LC1 Lrfq .9999
k6 LC2 Lrfq .9999
k7 LD1 Lrfq .9999
k8 LD2 Lrfq .9999

* connections to hybrids
Xa b13 ra1 ra2 la1 la2 hybrid
Xb b13 rb1 rb2 lb1 lb2 hybrid
Xc b23 rc1 rc2 lc1 lc2 hybrid
Xd b23 rd1 rd2 ld1 ld2 hybrid

.end

